# Maintainer: Daniel M. Capella <polyzen@archlinux.org>

pkgname=vivid
pkgver=0.9.0
pkgrel=1
pkgdesc='LS_COLORS manager with multiple themes'
arch=('loong64' 'x86_64')
url=https://github.com/sharkdp/vivid
license=('Apache' 'MIT')
depends=('gcc-libs')
makedepends=('rust')
source=("$url/archive/v$pkgver/vivid-$pkgver.tar.gz")
b2sums=('6bae858a27d704ff73ded2560ae74f6821c97517c8e8e3f9058619cfdf11bd3bd626074978951adb07cd799dc032a509f7e0c7171d8c7c0996d1f423bd9d7bf8')

prepare() {
  cd vivid-$pkgver
  cargo fetch --locked --target "$CARCH-unknown-linux-gnu"
}

build() {
  cd vivid-$pkgver
  cargo build --release --locked --offline
}

package() {
  cd vivid-$pkgver
  install -Dt "$pkgdir"/usr/bin target/release/vivid
  install -Dm644 -t "$pkgdir"/usr/share/licenses/vivid LICENSE-MIT
}
